

# NAG Library Routine Document

## F08KSF (ZGEBRD)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F08KSF (ZGEBRD) reduces a complex  $m$  by  $n$  matrix to bidiagonal form.

### 2 Specification

```
SUBROUTINE F08KSF (M, N, A, LDA, D, E, TAUQ, TAUP, WORK, LWORK, INFO)
INTEGER          M, N, LDA, LWORK, INFO
REAL (KIND=nag_wp)  D(*), E(*)
COMPLEX (KIND=nag_wp) A(LDA,*), TAUQ(*), TAUP(*), WORK(max(1,LWORK))
```

The routine may be called by its LAPACK name *zgebrd*.

### 3 Description

F08KSF (ZGEBRD) reduces a complex  $m$  by  $n$  matrix  $A$  to real bidiagonal form  $B$  by a unitary transformation:  $A = QBPH$ , where  $Q$  and  $P^H$  are unitary matrices of order  $m$  and  $n$  respectively.

If  $m \geq n$ , the reduction is given by:

$$A = Q \begin{pmatrix} B_1 \\ 0 \end{pmatrix} P^H = Q_1 B_1 P^H,$$

where  $B_1$  is a real  $n$  by  $n$  upper bidiagonal matrix and  $Q_1$  consists of the first  $n$  columns of  $Q$ .

If  $m < n$ , the reduction is given by

$$A = Q (B_1 \ 0) P^H = Q B_1 P_1^H,$$

where  $B_1$  is a real  $m$  by  $m$  lower bidiagonal matrix and  $P_1^H$  consists of the first  $m$  rows of  $P^H$ .

The unitary matrices  $Q$  and  $P$  are not formed explicitly but are represented as products of elementary reflectors (see the F08 Chapter Introduction for details). Routines are provided to work with  $Q$  and  $P$  in this representation (see Section 9).

### 4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

### 5 Arguments

- 1: M – INTEGER *Input*  
*On entry:*  $m$ , the number of rows of the matrix  $A$ .  
*Constraint:*  $M \geq 0$ .
- 2: N – INTEGER *Input*  
*On entry:*  $n$ , the number of columns of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .

- 3: A(LDA,\*) – COMPLEX (KIND=nag\_wp) array Input/Output  
**Note:** the second dimension of the array A must be at least  $\max(1, N)$ .  
*On entry:* the  $m$  by  $n$  matrix  $A$ .  
*On exit:* if  $m \geq n$ , the diagonal and first superdiagonal are overwritten by the upper bidiagonal matrix  $B$ , elements below the diagonal are overwritten by details of the unitary matrix  $Q$  and elements above the first superdiagonal are overwritten by details of the unitary matrix  $P$ .  
 If  $m < n$ , the diagonal and first subdiagonal are overwritten by the lower bidiagonal matrix  $B$ , elements below the first subdiagonal are overwritten by details of the unitary matrix  $Q$  and elements above the diagonal are overwritten by details of the unitary matrix  $P$ .
- 4: LDA – INTEGER Input  
*On entry:* the first dimension of the array A as declared in the (sub)program from which F08KSF (ZGEBRD) is called.  
*Constraint:*  $LDA \geq \max(1, M)$ .
- 5: D(\*) – REAL (KIND=nag\_wp) array Output  
**Note:** the dimension of the array D must be at least  $\max(1, \min(M, N))$ .  
*On exit:* the diagonal elements of the bidiagonal matrix  $B$ .
- 6: E(\*) – REAL (KIND=nag\_wp) array Output  
**Note:** the dimension of the array E must be at least  $\max(1, \min(M, N) - 1)$ .  
*On exit:* the off-diagonal elements of the bidiagonal matrix  $B$ .
- 7: TAUQ(\*) – COMPLEX (KIND=nag\_wp) array Output  
**Note:** the dimension of the array TAUQ must be at least  $\max(1, \min(M, N))$ .  
*On exit:* further details of the unitary matrix  $Q$ .
- 8: TAUP(\*) – COMPLEX (KIND=nag\_wp) array Output  
**Note:** the dimension of the array TAUP must be at least  $\max(1, \min(M, N))$ .  
*On exit:* further details of the unitary matrix  $P$ .
- 9: WORK(max(1, LWORK)) – COMPLEX (KIND=nag\_wp) array Workspace  
*On exit:* if INFO = 0, the real part of WORK(1) contains the minimum value of LWORK required for optimal performance.
- 10: LWORK – INTEGER Input  
*On entry:* the dimension of the array WORK as declared in the (sub)program from which F08KSF (ZGEBRD) is called.  
 If LWORK = -1, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.  
*Suggested value:* for optimal performance,  $LWORK \geq (M + N) \times nb$ , where  $nb$  is the optimal **block size**.  
*Constraint:*  $LWORK \geq \max(1, M, N)$  or LWORK = -1.
- 11: INFO – INTEGER Output  
*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO =  $-i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

## 7 Accuracy

The computed bidiagonal form  $B$  satisfies  $QBP^H = A + E$ , where

$$\|E\|_2 \leq c(n)\epsilon\|A\|_2,$$

$c(n)$  is a modestly increasing function of  $n$ , and  $\epsilon$  is the *machine precision*.

The elements of  $B$  themselves may be sensitive to small perturbations in  $A$  or to rounding errors in the computation, but this does not affect the stability of the singular values and vectors.

## 8 Parallelism and Performance

F08KSF (ZGEBRD) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08KSF (ZGEBRD) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The total number of real floating-point operations is approximately  $16n^2(3m - n)/3$  if  $m \geq n$  or  $16m^2(3n - m)/3$  if  $m < n$ .

If  $m \gg n$ , it can be more efficient to first call F08ASF (ZGEQRF) to perform a  $QR$  factorization of  $A$ , and then to call F08KSF (ZGEBRD) to reduce the factor  $R$  to bidiagonal form. This requires approximately  $8n^2(m + n)$  floating-point operations.

If  $m \ll n$ , it can be more efficient to first call F08AVF (ZGELQF) to perform an  $LQ$  factorization of  $A$ , and then to call F08KSF (ZGEBRD) to reduce the factor  $L$  to bidiagonal form. This requires approximately  $8m^2(m + n)$  operations.

To form the unitary matrices  $P^H$  and/or  $Q$  F08KSF (ZGEBRD) may be followed by calls to F08KTF (ZUNGBR):

to form the  $m$  by  $m$  unitary matrix  $Q$

```
CALL ZUNGBR('Q',M,M,N,A,LDA,TAUQ,WORK,LWORK,INFO)
```

but note that the second dimension of the array A must be at least M, which may be larger than was required by F08KSF (ZGEBRD);

to form the  $n$  by  $n$  unitary matrix  $P^H$

```
CALL ZUNGBR('P',N,N,M,A,LDA,TAUP,WORK,LWORK,INFO)
```

but note that the first dimension of the array A, specified by the argument LDA, must be at least N, which may be larger than was required by F08KSF (ZGEBRD).

To apply  $Q$  or  $P$  to a complex rectangular matrix  $C$ , F08KSF (ZGEBRD) may be followed by a call to F08KUF (ZUNMBR).

The real analogue of this routine is F08KEF (DGEBRD).

## 10 Example

This example reduces the matrix  $A$  to bidiagonal form, where

$$A = \begin{pmatrix} 0.96 - 0.81i & -0.03 + 0.96i & -0.91 + 2.06i & -0.05 + 0.41i \\ -0.98 + 1.98i & -1.20 + 0.19i & -0.66 + 0.42i & -0.81 + 0.56i \\ 0.62 - 0.46i & 1.01 + 0.02i & 0.63 - 0.17i & -1.11 + 0.60i \\ -0.37 + 0.38i & 0.19 - 0.54i & -0.98 - 0.36i & 0.22 - 0.20i \\ 0.83 + 0.51i & 0.20 + 0.01i & -0.17 - 0.46i & 1.47 + 1.59i \\ 1.08 - 0.28i & 0.20 - 0.12i & -0.07 + 1.23i & 0.26 + 0.26i \end{pmatrix}.$$

### 10.1 Program Text

```

Program f08ksfe

!      F08KSF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: nag_wp, zgebrd
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Integer                     :: i, info, lda, lwork, m, n
!      .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: a(:,,:), taup(:), tauq(:), work(:)
      Real (Kind=nag_wp), Allocatable   :: d(:), e(:)
!      .. Intrinsic Procedures ..
      Intrinsic                   :: min
!      .. Executable Statements ..
      Write (nout,*) 'F08KSF Example Program Results'
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) m, n
      lda = m
      lwork = 64*(m+n)
      Allocate (a(lda,n),taup(n),tauq(n),work(lwork),d(n),e(n-1))

!      Read A from data file

      Read (nin,*)(a(i,1:n),i=1,m)

!      Reduce A to bidiagonal form

!      The NAG name equivalent of zgebrd is f08ksf
      Call zgebrd(m,n,a,lda,d,e,tauq,taup,work,lwork,info)

!      Print bidiagonal form

      Write (nout,*)
      Write (nout,*) 'Diagonal'
      Write (nout,99999) d(1:min(m,n))
      If (m>=n) Then
         Write (nout,*) 'Superdiagonal'
      Else
         Write (nout,*) 'Subdiagonal'
      End If
      Write (nout,99999) e(1:min(m,n)-1)

99999 Format (1X,8F9.4)
End Program f08ksfe

```

## 10.2 Program Data

F08KSF Example Program Data

```
6 4 :Values of M and N
( 0.96,-0.81) (-0.03, 0.96) (-0.91, 2.06) (-0.05, 0.41)
(-0.98, 1.98) (-1.20, 0.19) (-0.66, 0.42) (-0.81, 0.56)
( 0.62,-0.46) ( 1.01, 0.02) ( 0.63,-0.17) (-1.11, 0.60)
(-0.37, 0.38) ( 0.19,-0.54) (-0.98,-0.36) ( 0.22,-0.20)
( 0.83, 0.51) ( 0.20, 0.01) (-0.17,-0.46) ( 1.47, 1.59)
( 1.08,-0.28) ( 0.20,-0.12) (-0.07, 1.23) ( 0.26, 0.26) :End of matrix A
```

## 10.3 Program Results

F08KSF Example Program Results

```
Diagonal
-3.0870  2.0660  1.8731  2.0022
Superdiagonal
 2.1126  1.2628 -1.6126
```

---